Solar Energy, Vol. 20, pp. 393–397. Pergamon Press 1978. Printed in Great Britain

CALCULATING THE POSITION OF THE SUN

ROBERT WALRAVEN
Department of Land, Air and Water Resources, University of California, Davis, CA 95616, U.S.A.

(*Received* 4 *June* 1977; *received for publication* 1 *September* 1977)

Abstract—The procedure for calculating the position of the sun by computer is discussed. The equations used to generate the values in The American Ephemeris and Nautical Almanac are presented in a simplified form that allows the position of the sun to be calculated rapidly to an accuracy of 0.01°.

INTRODUCTION

Increased emphasis on the use of solar energy has led to several applications where the position of the sun must be found accurately. For example, the mirrors of a solar tower generator should be positioned with an accuracy of about 0.05°[1]. The American Ephemeris and Nautical Almanac contains tables for determining the position of the sun, but these tables are not of much use for live-time computer tracking of the sun. Furthermore, the algorithm for generating the tabular values[2] is very complicated because the values are calculated to an angular resolution of 0.1 second of arc. To achieve this kind of accuracy, detailed perturbations of the earth's orbit by the moon and planets, and other small effects such as luni-solar precession and nutation of the earth's orbit, stellar aberration, refraction by the atmosphere, and parallax are considered.

The algorithm for generating The American Ephemeris and Nautical Almanac solar tables can be greatly simplified if many of the minor effects are neglected. A simplified algorithm will be presented below which can easily be implemented on a minicomputer or microcomputer for rapidly computing the position of the sun to within 0.01°. This algorithm has been used successfully for several years at Davis to position a polarizing radiometer precisely in live-time with respect to the sun by minicomputer.

CONCEPTS

The equatorial plane of the earth makes an angle of approximately 23.5° with respect to the plane of the orbit of the earth about the sun, as shown in Fig. 1. It is this tilt that causes the elevation of the sun at noon to vary throughout the year, giving rise to seasons, and complicating the computation of the sun's position at a particular time. Figure 1 could be used to help us visualize how we might determine where the sun is in the sky. However, it is not the most natural starting point for a visualization of the problem because as we watch the sky we tend to think of the earth fixed beneath our feet, and the sun, moon, planets and stars sweeping across the sky far away. Let us imagine, therefore, a *celestial sphere* of virtually infinite radius with the earth fixed at its center, as shown in Fig. 2. Rather than have the earth rotate about its axis, let the celestial sphere rotate about the earth.

The intersection of the equatorial plane of the earth with the celestial sphere is called the *celestial equator*. The motion of the earth about the sun shown in Fig. 1 is represented on the celestial sphere of Fig. 2 by a circular orbit, called the *ecliptic*, which is tilted at approximately 23.5° with respect to the celestial equator. The sun thus makes one revolution along the ecliptic per year in the direction opposite to the direction of rotation of the celestial sphere due to the motion of the earth about the sun. (Note that the celestial sphere makes one more revolution per year about the earth than the sun does.)

The sun crosses the celestial equator at vernal equinox on 21 March and at autumnal equinox on 23 September. The sun is at its greatest distance from the celestial equator at summer solstice on 21 June and at winter solstice on 22 December. The great circle that passes



Fig. 1. Orbit of the earth around the sun. The tilt of the earth with respect to the orbit is approximately 23.5°.



Fig. 2. The celestial sphere.

through the celestial north pole and vernal equinox is called the *hour circle of vernal equinox.*

TIME

In generating the American Ephemeris and Nautical Almanac solar tables, time is reckoned from 1900, 0 January, Greenwich Mean Noon. Many small correction terms to the quantities that enter the calculations may be dropped if time is reckoned from a date closer to the present date. In the discussion below, time is, therefore, reckoned in days from 1980, 1 January, Greenwich Mean Noon. The following algorithm will give the correct time for use in equations presented later:

$$\begin{cases} \Delta = \text{YEAR} - 1980 \\ \text{LEAP} = \Delta/4\ (\text{MOD}4) \\ \text{TIME} = \Delta{*}365 + \text{LEAP} + \text{DAY} - 1 + (\text{HOUR} \\ \qquad + (\text{MINUTE} + \text{SECOND}/60)/60)/24 \\ \text{IF}(\Delta = \text{LEAP}{*}4)\ \text{TIME} = \text{TIME} - 1 \\ \text{IF}((\Delta < 0) \cdot \text{AND} \cdot (\Delta \neq \text{LEAP}{*}4))\ \text{TIME} = \text{TIME} - 1 \end{cases}$$

where DAY is the day number of the year starting with 1 for 1 January, except in leap years when 1 should be subtracted from the day number if it is before 1 March.

LONGITUDE OF THE SUN

The position of the sun on the celestial sphere can be found by specifying only one parameter, the longitude L, as shown in Fig. 2. Since the earth's orbit about the sun is not circular, the speed of travel of the sun on the ecliptic varies slightly throughout the year, so the expression for the longitude is somewhat complicated. The longitude of the sun, in radians, is calculated as follows:

$$\begin{aligned} \theta &= 2\pi * \text{TIME}/365.25 \\ g &= -0.031271 - (4.53963 \times 10^{-7}) * \text{TIME} + \theta \\ L &= 4.900968 + (3.67474 \times 10^{-7}) * \text{TIME} \\ &\quad + (0.033434 - 2.3 \times 10^{-9} * \text{TIME}) * \sin g \\ &\quad + 0.000349 * \sin 2g + \theta. \end{aligned}$$

(The quantity g is the *mean anomaly of the earth.*)

RIGHT ASCENSION AND DECLINATION

An alternate way of expressing the position of the sun on the celestial sphere is in terms of its *right ascension* α and *declination* δ shown in Fig. 3(a). Right ascension is the angle between the hour circle of vernal equinox and the hour circle of the sun (the great circle through the celestial north pole and the sun) measured from the vernal equinox in an easterly direction. Declination is the angular distance from the sun to the celestial equator along the hour circle of the sun. Declination is positive if the sun is in the northern hemisphere of the celestial sphere, and negative if the sun is in the southern hemisphere.

It is easily shown by spherical trigonometry that the right ascension and declination of the sun are related to the longitude of the sun by

$$\tan \alpha = \cos \epsilon \tan L$$

$$\sin \delta = \sin \epsilon \sin L$$

where ϵ is the angle between the plane of the ecliptic and the plane of the celestial equator. The precise value of ϵ is

$$\epsilon = 23^\circ.4420 - (3^\circ.56 \times 10^{-7}) * \text{TIME}.$$

SIDERIAL TIME

To determine the position of the sun in the sky at a given time, it is necessary to know not only the position of the sun on the celestial sphere, but the position of the celestial sphere with respect to the earth at that time. The rotation of the celestial sphere is measured in siderial time relative to the local celestial meridian. The local celestial meridian, as shown in Fig. 3, is the great circle through the celestial poles and zenith (the point on the celestial sphere directly overhead for an observer on the earth). A siderial day is the time between two suc-



Fig. 3. (a) The celestial sphere showing the definitions of δ, S, H and α described in the text; (b) A spherical triangle from the celestial sphere defining the zenith distance Z and the azimuthal angle A. ϕ is the local latitude.

cessive transits of the hour circle of vernal equinox across the celestial meridian. A siderial clock reads zero at the instant the hour circle of vernal equinox crosses the local celestial meridian. Thus, locations on the earth with different longitudes will have different siderial times at a given instant.

The siderial time in hours for Greenwich, England at 0 hr Greenwich Universal Time (GUT) is given by

$$\mathrm{ST} = 6.720165 + 24 * (\mathrm{TIME}/365.25 - (\mathrm{YEAR} - 1980)) + 0.000001411 * \mathrm{TIME}.$$

The local siderial time at 0 hr GUT is obtained by subtracting from ST the local longitude LONG in hours west of Greenwich (1 hr = 15°).

We saw before that the celestial sphere rotates about the earth slightly faster than the sun. In fact,

$$\text{one standard day} = 1.0027379 \text{ siderial day}.$$

Furthermore, Local Standard Time (LST) is related to GUT by

$$\mathrm{GUT} = \mathrm{LST} + \mathrm{ZONE} - C$$

where ZONE is the international zone time given in Table 1, and C is zero unless daylight-savings time is in effect, in which case it is equal to one. The local siderial time at any LST is therefore equal to the local siderial time at 0 hr GUT plus 1.0027379 * (LST + ZONE − C).

Combining the effect of local longitude and LST gives the following expression for local siderial time:

$$S = \mathrm{ST} - \mathrm{LONG} + 1.0027379 * (\mathrm{LST} + \mathrm{ZONE} - C).$$

Table 1. The International zone times for several time zones

Time zone	Zone time (hr)
Atlantic Standard Time	4
Eastern Standard Time	5
Central Standard Time	6
Mountain Standard Time	7
Pacific Standard Time	8
Hawaii Standard Time	10

LOCAL AZIMUTH AND ELEVATION

Given the local sidereal time, the right ascension and declination of the sun, and the local latitude, then the local azimuth and elevation of the sun can be determined from spherical trigonometry. The local azimuthal angle A and the zenith distance Z are defined as shown in Fig. 3(b). The local azimuthal angle of the sun is the angle between the local celestial meridian and the line between the zenith and the sun. By convention, the local azimuthal angle is positive when the sun is east or south. The zenith distance Z is the angular distance from the zenith to the sun. The local elevation E of the sun is thus 90° minus the zenith distance,

$$E = 90° - Z.$$

The desired expression for the local azimuth and elevation follows directly from the spherical triangle shown in Fig. 3(b), where ϕ is the local latitude (that is, the angular distance from the zenith to the celestial equator), and the hour angle, H, is simply the right ascension minus the local siderial time, as shown in Fig. 3(a)

$$H = \alpha - S.$$

From spherical trigonometry the triangle in Fig. 3(b) yields

$$\cos Z = \sin \phi \sin \delta + \cos \phi \cos \delta \cos H$$

$$\sin A = \cos \delta \sin H/\sin Z.$$

A problem arises when A is computed. The arcsin function that is used in computers and calculators returns a value that is by convention in the range $-90°$ to $+90°$. If A is outside this range it must be corrected. As the zenith distance shown in Fig. 3(b) is increased, the azimuthal angle A will move from the region $A < 90°$ to the region $A > 90°$. The value of θ_0 for which $A = 90°$ can be computed from spherical trigonometry using the identity

$$\cos(90° - \delta) = \cos(90° - \phi)\cos Z + \sin(90° - \phi)\sin Z \cos(90° - A).$$

For $A - 90°$, this gives

$$\cos Z = \sin\delta/\sin\phi.$$

The algorithm for correcting A is therefore

$$\text{IF} \cos Z < \sin\delta/\sin\phi \text{ THEN} \begin{cases} \text{IF } A < 0 \text{ THEN } A = A + 360° \\ A = 180° - A. \end{cases}$$

Notice that if the solar declination is greater than the local altitude, this correction must be made for all A.

A FORTRAN program to calculate the local azimuth and elevation of the sun using the method presented in this paper and some typical calculations are given in Appendix A.

SUNRISE AND SUNSET

Sunrise and sunset do not occur exactly at the time when the elevation of the sun is 0° due to atmospheric diffraction and curvature of the earth. The elevation of the sun at sunrise or sunset is given approximately by[3]

$$E_0 = -0°.833 - 0.0214h^{1/2}$$

where h is the local height above sea level in feet.

REFERENCES

1. L. L. Vant-Hult and A. F. Hildebrandt, Solar thermal power system based on optical transmission. *Solar Energy* **18** (1976).
2. Newcomb, Tables of the sun. *Astronomical Papers prepared for the use of the American Ephemeris and Nautical Almanac*, Vol. 6 (1898).
3. *Explanatory Supplement to The Astronomical Ephemeris and The American Ephemeris and Nautical Almanac.* Her Majesty's Stationary Office, London (1961).

APPENDIX A

```
      SUBROUTINE SUNAE (YEAR,DAY,HR,MIN,SEC,ZONE,DASVTM,LAT,LONG,A,E)
C
C     THIS SUBROUTINE CALCULATES THE LOCAL AZIMUTH AND ELEVATION OF
C     THE SUN AT A SPECIFIED LOCATION AND TIME USING AN APPROXIMATION
C     TO THE EQUATIONS USED TO GENERATE THE NAUTICAL ALMANAC.
C
C     INPUT PARAMETERS
C       YEAR    = THE YEAR NUMBER (E.G., 1977).
C       DAY     = THE DAY NUMBER OF THE YEAR STARTING WITH 1 FOR
C                 JANUARY 1, EXCEPT IN LEAP YEARS WHEN 1 SHOULD BE
C                 SUBTRACTED FROM THE DAY NUMBER BEFORE MARCH 1.
C       HR,MIN,SEC = THE TIME OF DAY.
C       ZONE    = THE LOCAL INTERNATIONAL ZONE TIME (E.G.: P.S.T. = 8).
C       DASVTM  = =1 IF DAYLIGHT SAVINGS TIME IN EFFECT, ELSE = 0.
C       LAT     = THE LOCAL LATITUDE IN DEGREES (NORTH IS POSITIVE).
C       LONG    = THE LOCAL LONGITUDE IN DEGREES WEST OF GREENWICH.
C
C     OUTPUT PARAMETERS
C       A       = AZIMUTHAL ANGLE OF THE SUN (POSITIVE IS EAST OF SOUTH)
C       E       = ELEVATION OF THE SUN.
C
C-----------------------------------------------------------------------
C
      REAL MIN, LAT, LONG
      DATA TWOPI, RAD /6.2831853, 0.017453293/
      DELYR = YEAR - 1980.
      LEAP = IFIX(DELYR/4.)
      T = HR + (MIN + SEC/60.) / 60.
      TIME = DELYR*365. + LEAP + DAY - 1. + T/24.
      IF (DELYR .EQ. LEAP*4.) TIME = TIME - 1.
      IF ((DELYR.LT.0.) .AND. (DELYR.NE.LEAP*4.)) TIME = TIME - 1.
      THETA = (360.*TIME/365.25) * RAD
      G = -0.031271 - 4.53963E-7*TIME + THETA
      EL = 4.900968+3.67474E-7*TIME+(0.033434-2.3E-9*TIME)*SIN(G)
     *     +0.000349*SIN(2.*G) + THETA
      EPS = 0.409140 - 6.2149E-9*TIME
      SEL = SIN(EL)
      A1 = SEL * COS(EPS)
      A2 = COS(EL)
      RA = ATAN2(A1,A2)
      IF (RA .LT. 0.) RA = RA + TWOPI
      DECL = ASIN(SEL*SIN(EPS))
      ST = 1.759335 + TWOPI*(TIME/365.25-DELYR) + 3.694E-7*TIME
      IF (ST .GE. TWOPI) ST = ST - TWOPI
```

APPENDIX A (*CONT*)

```
      S = ST - LONG*RAD + 1.0027379*(ZONE-DASVTM+T)*15.*RAD
      IF (S .GE. TWOPI) S = S - TWOPI
      H = RA - S
      PHI = LAT * RAD
      E = ASIN(SIN(PHI)*SIN(DECL)+COS(PHI)*COS(DECL)*COS(H))
      A = ASIN(COS(DECL)*SIN(H)/COS(E))/RAD
      IF (SIN(E) .GE. SIN(DECL)/SIN(PHI)) GO TO 10
      IF (A .LT. 0.) A = A + 360.
      A = 180. - A
   10 E = E / RAD
      RETURN
      END
```

P O S I T I O N   O F   T H E   S U N

YEAR: 1977
DAY: 120 (APRIL 30)
LATITUDE = 38.538 DEGREES NORTH
LONGITUDE = 121.758 DEGREES WEST OF GREENWICH

HOUR	AZIMUTH	ELEVATION
1 AM	-179.10	-36.77
2 AM	163.02	-35.11
3 AM	146.79	-30.09
4 AM	132.94	-22.48
5 AM	121.26	-13.08
6 AM	111.15	-2.52
7 AM	101.94	8.76
8 AM	92.94	20.42
9 AM	83.36	32.17
10 AM	72.01	43.67
11 AM	56.75	54.30
12 PM	33.78	62.73
1 PM	.57	66.30
2 PM	-32.90	62.96
3 PM	-56.23	54.64
4 PM	-71.68	44.07
5 PM	-83.14	32.60
6 PM	-92.76	20.87
7 PM	-101.78	9.21
8 PM	-110.99	-2.06
9 PM	-121.07	-12.62
10 PM	-132.69	-22.02
11 PM	-146.45	-29.65
12 PM	-162.56	-34.72

Resumen—Es discutido un procedimiento para calcular la posición del Sol por computadora. Se presenta en una forma simplificada las ecuaciones usadas para generar los valores en "The Américan Ephemeris" y en el "Nautical Almanac" la que permite calcular rapidamente la posición del Sol con una precisión de 0,01 grado.

Résumé—On traite ici de la marche à suivre pour calculer la position du soleil par ordinateur. Les équations utilisées à la mise à jour des données dans l'Ephéméride Américain et l'Almanach Nautique sont présentées sous une forme simplifiée que permet de calculer rapidement la position du soleil à 0,01 degré près.